

Educational use of Maple in Engineering Mathematics

Harald Pleym

Telemark University College, Department of Technology

Kjolnes Ring 56, 3914 Porsgrunn, Norway

harald.pleym@hit.no

Abstract

The main objective of this paper is to show how the computer algebra system Maple can be integrated as a valuable and powerful tool into the learning process of engineering mathematics. Maple can be used in several ways:

- Only as a solving engine. That means, you enter a mathematical expression which Maple subsequently execute, and you are freed from tedious computations
- To present a method as a sequence of steps in much the same way as in a textbook. The primary purpose then is to present intermediate steps to show how to solve the problem. This is relevant in most of a student's homework solutions

In this paper these methods of using Maple will be demonstrated with selected examples from calculus, differential equations, integral transforms and linear algebra. One of the beautiful qualities of Maple is that much can be done with relatively few commands and without involving formal programming with Maple. But when intermediate steps in solving a problem is clear, and especially if the focus is to demonstrate what happens to a solution when parameters are changed, it is often desirable and also an easy task to put the commands in question between "proc" and "end" to make a procedure. Examples of mixing steps together in this way will be shown.

Introduction

At Telemark College, Department of Technology in Porsgrunn, Norway, we have since 1995 incorporated the computer algebra system Maple as an integrated part of the learning process in engineering mathematics. I have developed interactive worksheets to present worked examples throughout all the subjects in the curriculum. They are designed to be used in conjunction with the textbook the students use in order to show how the computer algebra system Maple can be integrated as a valuable and powerful tool into the learning process of the material. These sample worksheets are also intended to be used as guides when the students prepare their own worksheets solutions to exercises and problems in mathematics or other homework where Maple is used. The main objective in this paper is therefore only to show a few examples on how Maple can be used both as an "instructor" and also to replace the use of pen and paper to present for instance a students homework as a live interactive Maple worksheet instead of a "dead" paper.

Integration by partials fractions

The first problem students often have before they can make use of the method of partial fractions is that they don't know how to find the factors of $Q(x)$ in the integrand of the form $\frac{P(x)}{Q(x)}$, where $P(x)$ and $Q(x)$ are polynomials in x and the degree of $P(x)$ is less than the degree of $Q(x)$. Secondly, they have to write down the integrand as a sum of simpler fractions corresponding to the factors of $Q(x)$ before they can find the constants by adding up the fractions in the decomposition, and then equating coefficients of like powers of x in the numerator of the sum with those in $P(x)$. Here the aim is to illustrate how we first can make direct use of Maple to find the factors and the sum of simpler fractions, before we proceed to present the method where all steps are shown. This is usually required in a student's homework solution. To illustrate how the decomposition is effected, we consider an integral where $Q(x)$ can be written as a combination of linear factors and irreducible quadratic factors.

Example 1

Find $\int \frac{x^3 - 4x - 1}{x^4 - 2x^3 + 2x^2 - 2x + 1} dx$ using partial fraction decomposition.

Solution

We define the integrand as the function

> **r:=x->(x^3-4*x-1)/(x^4-2*x^3+2*x^2-2*x+1);**

R(x)=r(x);

$$R(x) = \frac{x^3 - 4x - 1}{x^4 - 2x^3 + 2x^2 - 2x + 1}$$

> **factor(%);**

$$R(x) = \frac{x^3 - 4x - 1}{(x^2 + 1)(x - 1)^2}$$

> **Rf:=rhs(%);**

The rational function $R(x)$ may be decomposed into the sum of simple fractions of the type

> **convert(%%,parfrac,x);**

$$R(x) = -2 \frac{1}{(x - 1)^2} + \frac{\frac{3}{2}}{x - 1} - \frac{1}{2} \frac{-5 + x}{x^2 + 1}$$

In the same way we could use Maple as an instructor to show us how the decomposition is effected when $Q(x)$ consists of **1) linear factors**, none of which are repeated, **2) linear factors**, all of which are the same, **3) Irreducible quadratic factors**, none of which are repeated, **4) Irreducible quadratic factors**, all of which are the same.

Now we decompose $R(x)$ into the sum of the simple fractions of the type

> **eq:=Rf=A/(x-1)^2+B/(x-1)+(C*x+D)/(x^2+1):=;**

$$\frac{x^3 - 4x - 1}{(x^2 + 1)(x - 1)^2} = \frac{A}{(x - 1)^2} + \frac{B}{x - 1} + \frac{Cx + D}{x^2 + 1}$$

We simplify the right-hand side

> **simplify(%);**

$$\frac{x^3 - 4x - 1}{(x^2 + 1)(x - 1)^2} = \frac{Ax^2 + A + Bx^3 - Bx^2 + Bx - B + Cx^3 - 2Cx^2 + Cx + Dx^2 - 2Dx + D}{(x^2 + 1)(x - 1)^2}$$

pick up the numerators

> **map(numer,%);**

$$x^3 - 4x - 1 =$$

$$Ax^2 + A + Bx^3 - Bx^2 + Bx - B + Cx^3 - 2Cx^2 + Cx + Dx^2 - 2Dx + D$$

and collect the coefficients of like powers.

> **eqa:=collect(%x):%;**

$$x^3 - 4x - 1 =$$

$$(B + C)x^3 + (-B + A + D - 2C)x^2 + (B - 2D + C)x + A + D - B$$

Equating the coefficients of x^3 , x^2 , x and the **constant term** on both sides gives

> **seq(map(coeff,eqa,x,i),i=0..3):%;**

$$-1 = A + D - B, -4 = B - 2D + C, 0 = -B + A + D - 2C, 1 = B + C$$

Thus we have four equations in A , B , C and D , which may be solved to give

> **solve({%},{A,B,C,D});**

$$\{A = -2, C = \frac{-1}{2}, B = \frac{3}{2}, D = \frac{5}{2}\}$$

Then we have

> **subs(% ,eq);**

$$\frac{x^3 - 4x - 1}{(x^2 + 1)(x - 1)^2} = -2 \frac{1}{(x - 1)^2} + \frac{\frac{3}{2}}{x - 1} + \frac{-\frac{1}{2}x + \frac{5}{2}}{x^2 + 1}$$

Now integration gives

> **map(Int,% ,x);**

$$\int \frac{x^3 - 4x - 1}{(x^2 + 1)(x - 1)^2} dx = \int \left(-2 \frac{1}{(x - 1)^2} + \frac{\frac{3}{2}}{x - 1} + \frac{-\frac{1}{2}x + \frac{5}{2}}{x^2 + 1} \right) dx$$

> **lhs(%)=value(rhs(%))+C;**

$$\int \frac{x^3 - 4x - 1}{(x^2 + 1)(x - 1)^2} dx = 2 \frac{1}{x - 1} + \frac{3}{2} \ln(x - 1) - \frac{1}{4} \ln(x^2 + 1) + \frac{5}{2} \arctan(x) + C$$

To check the result, we differentiate both sides

> **diff(% ,x);**

$$\frac{x^3 - 4x - 1}{(x^2 + 1)(x - 1)^2} = -2 \frac{1}{(x - 1)^2} + \frac{\frac{3}{2}}{x - 1} - \frac{1}{2} \frac{x}{x^2 + 1} + \frac{\frac{5}{2}}{x^2 + 1}$$

> **simplify(%);**

$$\frac{x^3 - 4x - 1}{(x^2 + 1)(x - 1)^2} = \frac{x^3 - 4x - 1}{(x^2 + 1)(x - 1)^2}$$

Newton's Method

Newton-Raphson's iterative method for generating a sequence x_1, x_2, x_3, \dots of approximations to a solution of $f(x) = 0$ is given by the formula

$$x_{n+1} = x_n - \frac{f(x_n)}{\frac{\partial}{\partial x} f(x_n)}$$

The formula permits us to go from the n th approximation x_n to the next approximation x_{n+1} . The formula is very easy to set up and apply repeatedly. When iterating for the root, it is often usual to present the calculation in tabular form. This can be done using the Maple spreadsheet, or calculate the values using a [for-loop](#) and present the result in a [sequence](#) of values or in an [array](#). So, the goal here is to demonstrate that when the intermediate steps in solving the problem is understood, we put the individual commands between "proc" and "end" to make a procedure and make use of this to execute Newton's method on a routine basis.

Example 2

Find the root of $f(x) = 4x^3 - 42x^2 + 60x + 15 = 0$ which lies between $x = 1$ and $x = 2$.

Solution

> **f:=x->4*x^3-42*x^2+60*x+15: 'f(x)'=f(x);**

$$f(x) = 4x^3 - 42x^2 + 60x + 15$$

We define Newton's formula

> **N:=x->x-f(x)/D(f)(x);**

$$N := x \rightarrow x - \frac{f(x)}{D(f)(x)}$$

Starting with $x = 1.1$, we get

> **N(1.1); N(%); N(%); N(%);**

3.085682327

2.117484062

1.988066154

1.983255914

If we use a [for loop](#), we can represent the calculations as a sequence of values.

> **x[0]:=1.1:**

> **for n from 1 to 6 do**

x[n]:=evalf(x[n-1]-f(x[n-1])/D(f)(x[n-1]));

od:

The **sequence of values** become

```
> S:=seq([n,x[n]],n=0..6):%;
```

```
[0, 1.1], [1, 3.085682327], [2, 2.117484062], [3, 1.988066154],  
[4, 1.983255914], [5, 1.983248839], [6, 1.983248839]
```

We can also show the values in an **array**

```
> ['n','x[n]']; array([S]);
```

$[n, x_n]$	
0	1.1
1	3.085682327
2	2.117484062
3	1.988066154
4	1.983255914
5	1.983248839
6	1.983248839

If we use Maple's **spreadsheet**, we can present the various steps in the calculations.

	A	B	C	D	E
1	n	x_n	$f(x_n)$	$\frac{\partial}{\partial x} f(x_n)$	$-\frac{f(x_n)}{\frac{\partial}{\partial x} f(x_n)}$
2	0	1.10000000	35.50400000	-17.88000000	1.98568232
3	1	3.08568232	-82.23884820	-84.94009040	-96819826
4	2	2.11748406	-8.29100250	-64.06379620	-.12941790
5	3	1.98806615	-.28653960	-59.56867250	-.00481023
6	4	1.98325591	-.00042020	-59.39384860	-.70748067 10 ⁻⁵
7	5	1.98324883	0.00000000	-59.39359100	0.00000000
8	6	1.98324883	0.00000000	-59.39359100	0.00000000

It is more convenient to make a small procedure for finding approximations by Newton's method on a **routine basis**, when the intermediate steps are known.

```
> Newton:=proc(f,x0,N)
```

```
  x[0]:=x0;
```

```
  for n from 1 to N do
```

```
    x[n]:=evalf(x[n-1]-f(x[n-1])/D(f)(x[n-1]));
```

```
  od:
```

```
  [seq(x[n],n=0..N)];
```

```
end:
```

```
> Newton(f,1.1,5);
```

```
[1.1, 3.085682327, 2.117484062, 1.988066154, 1.983255914, 1.983248839]
```

First order linear differential equation - use of an integrating factor

Maple's `dsolve` command for solving differential equation, and in this context a first-order linear equation, usually gives us the answer promptly. But from an educational point of view it is important to use standard techniques and go through the solution process step by step as the textbooks do. When we use Maple, this step-by-step solution can be presented in a dynamic and interactive form. The following example shows this approach.

Example 3

a) Find the particular solution of the *RL* circuit given by $20 \frac{di}{dt} + 40 i = 200 \sin(t)$ and satisfying the initial condition $i(0) = 0$.

Solution

> **restart:**

> **20*diff(i(t),t)+40*i(t)=200*sin(t):%**

$$20 \left(\frac{\partial}{\partial t} i(t) \right) + 40 i(t) = 200 \sin(t)$$

Division of both sides of the equation by 20 gives

> **deq:=%/20:%;**

$$\left(\frac{\partial}{\partial t} i(t) \right) + 2 i(t) = 10 \sin(t)$$

Multiplication both sides by the integrating factor

> **rho=exp(Int(2,t));**

$$\rho = e^{\int 2 dt}$$

> **value(%); rho:=rhs(%):**

$$\rho = e^{(2t)}$$

gives

> **simplify(expand(deq*rho));**

$$e^{(2t)} \left(\frac{\partial}{\partial t} i(t) \right) + 2 e^{(2t)} i(t) = 10 e^{(2t)} \sin(t)$$

Here we see that the left-hand side of this equation may be written as the derivative of the product $e^{(2t)} i(t)$. Thus we have

> **eq:=Diff(rho*i(t),t)=rhs(%):%;**

$$\frac{\partial}{\partial t} e^{(2t)} i(t) = 10 e^{(2t)} \sin(t)$$

You should always check your work at this point by differentiating the left-hand side of the above equation.

> **lhs(%)=value(lhs(%));**

$$\frac{\partial}{\partial t} e^{(2t)} i(t) = e^{(2t)} \left(\frac{\partial}{\partial t} i(t) \right) + 2 e^{(2t)} i(t)$$

We integrate both sides of **eq** to obtain

> **map(Int,eq,t);**

$$\int \frac{\partial}{\partial t} e^{(2t)} i(t) dt = \int 10 e^{(2t)} \sin(t) dt$$

> **map(int,eq,t)+(0=C);**

$$e^{(2t)} i(t) = -2 e^{(2t)} \cos(t) + 4 e^{(2t)} \sin(t) + C$$

> **simplify(isolate(%,i(t))):%;**

$$i(t) = (-2 e^{(2t)} \cos(t) + 4 e^{(2t)} \sin(t) + C) e^{(-2t)}$$

> **eq1:=expand(%,power):%;**

$$i(t) = -2 \cos(t) + 4 \sin(t) + \frac{C}{(e^t)^2}$$

Substituting the initial conditions

> **t=0,i(0)=0;**

$$t = 0, i(0) = 0$$

gives

> **subs(%,eq1);**

$$0 = -2 \cos(0) + 4 \sin(0) + \frac{C}{(e^0)^2}$$

> **isolate(%,C);**

$$C = 2$$

Then we have the particular solution

> **subs(%,eq1);**

$$i(t) = -2 \cos(t) + 4 \sin(t) + \frac{2}{(e^t)^2}$$

> **lhs(%)=map(simplify,rhs(%),power);**

$$i(t) = -2 \cos(t) + 4 \sin(t) + 2 e^{(-2t)}$$

It is recommended to check the answer.

> **subs(%,lhs(deq)-rhs(deq)=0);**

$$\left(\frac{\partial}{\partial t} (-2 \cos(t) + 4 \sin(t) + 2 e^{(-2t)}) \right) - 4 \cos(t) - 2 \sin(t) + 4 e^{(-2t)} = 0$$

> **%;**

$$0 = 0$$

The Laplace transform

Laplace transforms are used frequently in the analysis and design of engineering systems. The usefulness of the transform is that a differential equation can be transformed into an algebraic equation which involves no derivatives at all. And so it reduces the problem of solving an initial value problem to the inspection of a table of transforms. But without the use of a computer algebra system (CAS), the method is practically only if a sufficient large table of transforms is available. The use of Maple enable us if necessary to add our own functions to Laplace's internal lookup table by using the [addtable](#) functionus. And we are also freed from tedious manual partial fraction decomposition and other algebraic

manipulations. The following step by step solution shows that Maple's Laplace transforms package is very useful for elegant and quick calculations of transforms, suitable for use in the classroom from a pedagogical point of view. It is convenient to start the solution process by defining the Laplace transform and the inverse Laplace transform by functions. The purpose of the [alias](#) facility is to allow us to state abbreviations for the longer names that Maple uses.

> **with(inttrans):**

> **L:=f->laplace(f,t,s):invL:=F->invlaplace(F, s, t): alias(X(s)=L(x(t)),Y(s)=L(y(t)), F(s)=L(f(t))):**

Example 4

Solve the system

$$2x'' = -6x + 2y,$$

$$y'' = 2x - 2y + 40 \sin(3t)$$

$$x(0) = 0, x'(0) = 0, y(0) = 0, y'(0) = 0$$

Solution

> **sys:=2*diff(x(t),t,t)=-6*x(t)+2*y(t), diff(y(t),t,t)=2*x(t)-2*y(t)+4*sin(3*t):%;**

$$2 \left(\frac{\partial^2}{\partial t^2} x(t) \right) = -6 x(t) + 2 y(t), \frac{\partial^2}{\partial t^2} y(t) = 2 x(t) - 2 y(t) + 4 \sin(3t)$$

Initial conditions are

> **inits:=x(0)=0,D(x)(0)=0,y(0)=0,D(y)(0)=0:%;**

$$x(0) = 0, D(x)(0) = 0, y(0) = 0, D(y)(0) = 0$$

Laplace transformation gives

> **L({sys});**

$$\{ 2 s (s X(s) - x(0)) - 2 D(x)(0) = -6 X(s) + 2 Y(s),$$

$$s (s Y(s) - y(0)) - D(y)(0) = 2 X(s) - 2 Y(s) + \frac{12}{s^2 + 9} \}$$

> **subs(inits,%);**

$$\{ s^2 Y(s) = 2 X(s) - 2 Y(s) + \frac{12}{s^2 + 9}, 2 s^2 X(s) = -6 X(s) + 2 Y(s) \}$$

> **solve(%,{X(s),Y(s)});**

$$\left\{ Y(s) = 12 \frac{s^2 + 3}{14 s^4 + 49 s^2 + 36 + s^6}, X(s) = 12 \frac{1}{14 s^4 + 49 s^2 + 36 + s^6} \right\}$$

The partial fraction decomposition gives

> **convert(% ,parfrac,s);**

$$\left\{ X(s) = \frac{3}{10} \frac{1}{s^2 + 9} - \frac{4}{5} \frac{1}{s^2 + 4} + \frac{1}{2} \frac{1}{s^2 + 1}, Y(s) = \frac{4}{5} \frac{1}{s^2 + 4} - \frac{9}{5} \frac{1}{s^2 + 9} + \frac{1}{s^2 + 1} \right\}$$

The inverse Laplace transform yields the solution

> **invL(%);**

$$\left\{ x(t) = \frac{1}{10} \sin(3t) - \frac{2}{5} \sin(2t) + \frac{1}{2} \sin(t), y(t) = \frac{2}{5} \sin(2t) + \sin(t) - \frac{3}{5} \sin(3t) \right\}$$

We can also check this result by direct solution.

> **dsolve({sys,inits},{x(t),y(t)},method=laplace);**

$$\left\{ x(t) = \frac{1}{10} \sin(3t) - \frac{2}{5} \sin(2t) + \frac{1}{2} \sin(t), y(t) = \frac{2}{5} \sin(2t) + \sin(t) - \frac{3}{5} \sin(3t) \right\}$$

Eigenvalues and Eigenvectors

Eigenvalue problems are among the most important problems in connection with matrices and engineering applications, and they provide critical information in engineering design.

The advantage of using Maple is that the method described in textbooks and used by instructors on the blackboard can be reproduced on the computer in the classroom.

Eigenvalues are the roots of the characteristic equation $|\lambda \mathbf{I} - \mathbf{A}| = 0$. Solving this equation is almost always easier said than done on a blackboard. And for large matrices we may not be able to solve the characteristic equation exactly. And in this case a numerical approach may be the only possibility, which can be done interactively in the classroom using the power of Maple. The example here shows the sequences of calculations of eigenvalues and the corresponding eigenvectors using the "blackboard" method.

Example 5

Find the eigenvalues and associated eigenvectors of the matrix

$$A = \begin{bmatrix} 3 & 0 & 0 \\ -4 & 6 & 2 \\ 16 & -15 & -5 \end{bmatrix}, \text{ and display the factorization } A = PDP^{-1}$$

Solution

> **with(linalg):**

> **A:=matrix(3,3,[3,0,0,-4,6,2,16,-15,-5]);**

$$A := \begin{bmatrix} 3 & 0 & 0 \\ -4 & 6 & 2 \\ 16 & -15 & -5 \end{bmatrix}$$

> **'det'(charmat(A,lambda))=charpoly(A,lambda);**

$$\det \left(\begin{bmatrix} \lambda - 3 & 0 & 0 \\ 4 & \lambda - 6 & -2 \\ -16 & 15 & \lambda + 5 \end{bmatrix} \right) = (\lambda - 3)(\lambda^2 - \lambda)$$

The characteristic equation of A is

> **rhs(%)=0;**

$$(\lambda - 3)(\lambda^2 - \lambda) = 0$$

> **lambda:=solve(%,lambda);**

$$\lambda := 0, 1, 3$$

With $\lambda_1 = 0$ we get

> **Ac:=charmat(A,lambda[1]): X:=Vector([x,y,z]): b:=Vector([0,0,0]):**

> **evalm(Ac)*X=b;**

$$\begin{bmatrix} -3 & 0 & 0 \\ 4 & -6 & -2 \\ -16 & 15 & 5 \end{bmatrix} \begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

The augmented coefficient matrix of the system is

> **augment(Ac,b);**

$$\begin{bmatrix} -3 & 0 & 0 & 0 \\ 4 & -6 & -2 & 0 \\ -16 & 15 & 5 & 0 \end{bmatrix}$$

Gauss-Jordan elimination with back substitution gives

> **gaussjrd(%);**

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & \frac{1}{3} & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

> **evalm(X)=backsub(%);**

$$[x, y, z] = \left[0, -\frac{1}{3}t_1, -t_1 \right]$$

The eigenvector

> **v[1]=subs(_t[1]=-3,convert(rhs(%),matrix));**

$$v_1 = \begin{bmatrix} 0 \\ 1 \\ -3 \end{bmatrix}$$

is associated with λ_1 to within a constant multiple. The eigenvectors for the remaining two eigenvalues can be produced in the same manner. To show the **complete solution**, we use the eigenvector command.

> **ev:=eigenvectors(A):%;**

$$[0, 1, \{[0, 1, -3]\}], \left[1, 1, \left\{ \left[0, 1, \frac{-5}{2} \right] \right\} \right], [3, 1, \{[1, 0, 2]\}]$$

The matrix P with the eigenvectors as columns is

> **P:=augment(seq(ev[i,3,1],i=1..3));**

$$P := \begin{bmatrix} 0 & 0 & 1 \\ 1 & 1 & 0 \\ -3 & \frac{-5}{2} & 2 \end{bmatrix}$$

> **unprotect(D):**

The diagonal entries of D are eigenvalues of A .

> **D:=diag(seq(ev[i,1],i=1..3));**

$$D := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

> **A=evalm(P)&*evalm(D)&*inverse(P);**

$$A = \left(\begin{bmatrix} 0 & 0 & 1 \\ 1 & 1 & 0 \\ -3 & \frac{-5}{2} & 2 \end{bmatrix} \&* \begin{bmatrix} 0 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 3 \end{bmatrix} \right) \&* \begin{bmatrix} 4 & -5 & -2 \\ -4 & 6 & 2 \\ 1 & 0 & 0 \end{bmatrix}$$

> **evalm(%);**

$$\begin{bmatrix} 3 & 0 & 0 \\ -4 & 6 & 2 \\ 16 & -15 & -5 \end{bmatrix} = \begin{bmatrix} 3 & 0 & 0 \\ -4 & 6 & 2 \\ 16 & -15 & -5 \end{bmatrix}$$

References

- [1] C. H. Edwards and D. E. Penney. *Calculus with Analytic Geometry*, Prentice-Hall International, Inc., 1998.
- [2] C. H. Edwards and D. E. Penney. *Differential Equations & Linear Algebra*, Prentice-Hall International, Inc., 2001.